

# 17. Archery

## *General Description*

This program is designed to demonstrate the effect of speed and gravity on the course of a moving object. The simulation is in the form of an archery contest. The instructions for the game appear on the screen when the program is started. The game requires for the user to set the angle of elevation (from 20 to 70), and the speed of the arrow's flight (which can be from 50 to 140). Points are awarded according to which part of the target (if any) is hit with each of your five arrows:- 30 for a bullseye, 20 for the middle ring and 10 for the outer edge.

Press any key to start.

## *Detailed Description*

**Lines 10-60** Credits.

**70-100** Set up variables.

**110-120** GOTO procedure for instructions.

**130-150** Setup colours, sounds and characters.

**160** Set D = distance and X = distance to plot target, also goes to procedure to display target.

**170-260** Display score, distance and asks for angle of trajectory, and speed in metres per second.

**270** Changes degrees to radians.

**280-320** Erases old arrow.

**330-360** Has arrow hit target?

**370-390** Display arrow in new position.

**440-430** Do it again until arrow hits something.

**440-540** Procedure to display target.

**550-590** Add to score.

**600-670** Display ratings.

**680-850** Procedure to display instructions.

### *Educational Note*

Whilst this simulation is primarily just for fun, its use does encourage users to take into account the consequences of two different variables on their arrows' flight, and witness the outcome of their decision before their very eyes! The game is probably best run with two players taking alternate goes.

NB For projectiles under gravity the following formula was used:

$$Y = X * \tan(a) - (X * X * G) / (2 * U * U * \cos(a) * \cos(a))$$

Where x=distance from target; U=initial velocity; a=angle of trajectory; G=9.81 (force of gravity); Y=height of projectile.

### *Program Listing*

```
>
10 REM *****
20 REM *           ARCHERY           *
30 REM *           =====           *
40 REM *           WRITTEN FOR THE BBC-A *
50 REM *           BY ANDREW PUSEY FEB 1983 *
60 REM *****
70 O=50
80 MG=9.8
90 SCORE=0
100 MODE7
110 PROC_instructions
120 MODE5:VDU23;8202;0;0;0;
130 VDU19,2,2,0,0,0
140 ENVELOPE 1,1,0,0,0,20,20,20,125,-1,0,0,100,100
150 VDU23,240,232,228,100,254,228,196,200,224
160 D=RND(10)+20:X=D*30:PROC_BOARD
170 FOR SHOT=1TO5
180   PRINT TAB(1,28);CHR$240
190   PRINT TAB(2,0);SPC(100);TAB(2,0);"DISTANCE="D
200   PRINT TAB(1,6);"SCORE=";SCORE
210   SOUND 1,0,0,0
220   INPUTTAB(2,1)"ANGLE  =" AN
230   IF AN<20 OR AN>70THENVDU7:GOTO190
240   INPUTTAB(2,2)"SPEED  =" U
250   PRINT TAB(0,0);SPC(255)
260   IF U<50 OR U>140THENVDU7:GOTO190
270   A=RAD(AN)
280   FOR X=150TO1279*(D/25)STEP8
290     Y=X*TAN(A)-(X^2*MG)/(2*U^COS(A)^2)
300     MOVE X/(D/25)-30,O+50
310     PLOT 6,X/(D/25),Y+50
320     SOUND 1,1,Y/2+5,1
330     C=POINT(X/(D/25)+9,Y+42)
340     IF C>0THEN550
```

```

350     E=POINT(X/(D/25)+9,Y+34)
360     IF E>0THENC=E:GOTO550
370     MOVE X/(D/25)-30,O+50
380     PLOT 6,X/(D/25),Y+50
390     IF C=-1ORE=-1THEN550
400     O=Y
410     IF Y<-10 THEN 590
420     NEXT
430     GOTO 590
440     DEFPROC_BOARD
450     MOVE X,100
460     GCOL0,3
470     DRAW X+150,250
480     GCOL0,2
490     MOVE X+30,130
500     DRAW X+120,220
510     GCOL0,1
520     MOVE X+60,160
530     DRAW X+90,190
540     ENDPROC
550     IF C=-1 THEN 590
560     SCORE=SCORE+(4-C)*10
570     SOUND1,-15,100,1:A$=INKEY$(100)
580     PRINT TAB(0,0);SPC(100)
590     NEXT SHOT
600     MODE6
610     SOUND1,0,0,0
620     PRINT TAB(4,4);"You scored "SCORE" points."
630     IF SCORE>120 PRINT'" Are you Robin Hood ??!":END
640     IF SCORE>90 PRINT'" Not bad - I can see you've been p
ractising.":END
650     IF SCORE>50 PRINT'" Not too bad, try getting a better
bow.":END
660     IF SCORE>20 PRINT'" If I was you, I would try taking
some lessons before you come here again !!!":END
670     PRINT"Try looking next time !!!";CHR$(7):END
680     DEFPROC_instructions
690     CLS
700     PRINTTAB(10,1);CHR$(141);CHR$(130);"ARCHERY"
710     PRINTTAB(10,2);CHR$(141);CHR$(130);"ARCHERY"
720     PRINTCHR$(131);" Welcome to the game of Archery."
730     PRINT'CHR$(131);"The idea of this game is to fire all"
740     PRINT'CHR$(131);"your arrows into the target."
750     PRINT'CHR$(131);"If you hit the white (outside) then"
760     PRINT'CHR$(131);"you score 10 points"
770     PRINT'CHR$(131);"the green 20 points"
780     PRINT'CHR$(131);"the red (Bullseye) 30 points."
790     PRINT'CHR$(131);" You start with 5 arrows so the"
800     PRINT'CHR$(131);"maximum score is 150."
810     PRINT'CHR$(131);"The angle is from 20 to 70."
820     PRINT'CHR$(131);"and the speed is from 50 to 140."
830     PRINTCHR$(136);CHR$(129);"Press any key to begin."
840     A$=GET$
850     ENDPROC

```

